2022/1/22(土), 3/21(月)　JAPLA 研究会/パレット柏/

Ｊによるアフィン変換を使った奇数魔方陣

西川　利男

はじめに

　これまで、奇数魔方陣[1],　偶数魔方陣[2]のＪプログラムについて報告してきた。
今回別に、アフィン変換を用いた魔方陣の作成方法[3]の知見を得たので、紹介する。
　いうまでもなく､魔方陣とは正方形に並べた自然数の配列において､タテ､ヨコ､ナナメの各和を同じにするには、どうしたらよいか、という古くからの問題である。
　アフィン変換とは何かについてはいずれ説明することになるが、まずは最も基本となる３次（３ｘ３）の魔方陣をＪで作ることからはじめる。
なお、しばらくぶりの JAPLA 研究会なので、Ｊの易しい入門もかねて説明したいと思う。

　[1]「Jプログラムにより「誰でもできる奇数次の魔方陣」をつくってみよう」
JAPLA 研究会資料 2018/8/4
　[2] 「Jプログラムにより「誰でもできる偶数次の魔方陣」をつくってみよう」
JAPLA 研究会資料　2018/9/8
　[3] 佐藤肇、一楽重雄、「幾何の魔術　魔方陣から現代数学へ」日本評論社(2000).

１．３ｘ３魔方陣

　Ｊなどプログラミング言語によっては、自然数を 1,2,3,.. ではなく、0,1,2, ...
(0－オリジン)のようにとる方式が少なくない。
　まず、0－オリジンで３つの数からなるベクトルＮを作る。

```
   N =: i. 3
   N
0 1 2
```

　これを元に３行３列から成る配列を次のようにして作る。

```
   M0 =: (3, 3)$N
  M0
0 1 2
0 1 2
0 1 2
```

一方、J ではベクトル N の各要素を右に望む数だけシフトしたベクトルを演算(|.)により作ることができる。最後の要素は右シフトすると回転として最初に入れられる。

```
   NR =: 1 |. N
   NR
1 2 0
   ML =: 2 |. N
   ML
2 0 1
```

これを使って、左右に１つずつずれた配列をつくる。

```
   MR =: (N,: (1 |. N)), (2 |. N)
   MR
0 1 2
1 2 0
2 0 1
  ML =: (N,: (2 |. N)), (1 |. N)
  ML
0 1 2
2 0 1
1 2 0
```

このような１つずつずれた配列を Latin 配列ということがある。

ここで、MR と ML とから、要素を１つずつとって組にした配列をつくる。そのためには J のボックス化機能(<)をランク０（"(0)）で用いて、つぎのようにつくることができる。

```
    <"(0) N
+-+-+-+
|0|1|2|
+-+-+-+
    <"(0) NR
+-+-+-+
|1|2|0|
+-+-+-+
    <"(0) NL
+-+-+-+
|2|0|1|
+-+-+-+
    <"(0) MR
```

```
+-+-+-+
|0|1|2|
+-+-+-+
|1|2|0|
+-+-+-+
|2|0|1|
+-+-+-+
   <"(0) ML
+-+-+-+
|0|1|2|
+-+-+-+
|2|0|1|
+-+-+-+
|1|2|0|
+-+-+-+
```

さらに、MR と ML との各ボックス内のそれぞれの要素をボックス内で(, L:0)で、J の動詞結合(,)により、結合させる。これをオイラー方陣と呼ぶ。

```
  MRL =: (<"(0) MR) , L:0 (<"(0) ML)
  NRL
+---+---+---+
|0 0|1 1|2 2|
+---+---+---+
|1 2|2 0|0 1|
+---+---+---+
|2 1|0 2|1 0|
+---+---+---+
```

同様にして

```
  MLR =: (<"(0) ML) , L:0 (<"(0) MR)
  MLR
+---+---+---+
|0 0|1 1|2 2|
+---+---+---+
|2 1|0 2|1 0|
+---+---+---+
|1 2|2 0|0 1|
+---+---+---+
```

次に各ボックス内の２つの値の並びを３進数と見なして、その値を(3 #. L:0)により示す。MLR については次のようになる。

```
   3 #. L:0 (< "(0) ML) ,L:0 (< "(0) MR)
+-+-+-+
|0|4|8|
+-+-+-+
|7|2|3|
+-+-+-+
|5|6|1|
+-+-+-+
```

ここで各値を０－オリジンから１－オリジンに戻す。実はここで出来た配列 Mahoujin が魔方陣になる。そして、これらの一連の操作がアフィン変換である。

```
   Mahoujin =: >: (L:0) 3 #. L:0 (< "(0) ML) ,L:0 (< "(0) MR)
+-+-+-+
|1|5|9|
+-+-+-+
|8|3|4|
+-+-+-+
|6|7|2|
+-+-+-+
```

ところが、タテ、ヨコについては一定の和になるが、ナナメについては定和にならない。そこで、最後の行を最初に移すと、ナナメについても一定和になる。

最後にボックスをはずした配列Magicが最終結果になる。

```
   Magic =: ((2{Mahoujin),: (0{Mahoujin)) , (1{Mahoujin)
   Magic
6 7 2
1 5 9
8 3 4
```

タテの和は +/ で、ヨコの和は +/”(0)で求めることができる。

```
   +/  Magic
15 15 15
   +/” (0) Magic
15 15 15
```

また次の操作diagにより対角要素を取り出すことができるので

```
   diag =: i. @ # {"_1 ]
   diag Magic
6 5 4
   +/ diag Magic
15
```

逆対角要素の和を調べるには、回転(|.)をランク１（"(1)）で用いて、次のように行う。

```
   |. "(1) Magic
2 7 6
9 5 1
4 3 8
   diag |. "(1) Magic
2 5 8
   +/ diag |. "(1) Magic
15
```

２．５ｘ５魔方陣

これまで、3 x 3 魔方陣で行った方法を

```
NN =: i.5
```

として、5 x 5 魔方陣を作成した。そのプログラムは以下のとおりである。

```
NB. 5*5 Magic Square ============================================
NN =: i.5
MM0 =: (5,5) $ NN
MMR =: ((NN ,: (1 |. NN)) , (2 |. NN) ,: (3 |. NN)) , (4 |. NN)
MML =: (((NN ,: (4 |. NN)) , (3 |. NN)), (2 |. NN)) , (1 |. NN)
MMahoujin =: (<"(0) MML) , L:0  (<"(0) MMR)
MMagic =: (5, 5) $ ; 5 #. L:0 MMahoujin
Magic55 =: 1 + ((3, 4) { MMagic) ,  (0, 1, 2) { MMagic
```

以下は実行結果のみを記す。前章での説明から、おわかりいただけると思う。Magic55 が最終結果である。

```
   NN
0 1 2 3 4
   MM0
0 1 2 3 4
0 1 2 3 4
0 1 2 3 4
0 1 2 3 4
0 1 2 3 4
   MMR
0 1 2 3 4
1 2 3 4 0
2 3 4 0 1
3 4 0 1 2
4 0 1 2 3
   MML
0 1 2 3 4
4 0 1 2 3
3 4 0 1 2
2 3 4 0 1
1 2 3 4 0
```

```
   MMahoujin
+---+---+---+---+---+
|0 0|1 1|2 2|3 3|4 4|
+---+---+---+---+---+
|4 1|0 2|1 3|2 4|3 0|
+---+---+---+---+---+
|3 2|4 3|0 4|1 0|2 1|
+---+---+---+---+---+
|2 3|3 4|4 0|0 1|1 2|
+---+---+---+---+---+
|1 4|2 0|3 1|4 2|0 3|
+---+---+---+---+---+
   Magic55
14 20 21  2  8
10 11 17 23  4
 1  7 13 19 25
22  3  9 15 16
18 24  5  6 12
   +/ Magic55
65 65 65 65 65
   +/"(1) Magic55
65 65 65 65 65
   diag Magic55
14 11 13 15 12
   +/ diag Magic55
65
   |. ("1) Magic55
 8  2 21 20 14
 4 23 17 11 10
25 19 13  7  1
16 15  9  3 22
12  6  5 24 18
   diag |. ("1) Magic55
8 23 13 3 18
   +/ diag |. ("1) Magic55
65
```

３．おわりに

３次と５次の奇数魔方陣をＪプログラムにより作成した。それでは同じ方法を用いて偶数魔方陣を作ることが出来るのだろうか？残念ながら、うまくいかない。

ここで本格的なアフィン変換の知識が必要となる。４次魔方陣から始めて、１０次魔方陣のＪプログラムについてはいずれ稿をあらためて紹介したいと思う。

```
NB. Afin_MagicSq.ijs
NB. Latin Matrix and Magic Square
NB. 「幾何の魔術」 佐藤肇 一楽重雄
NB.  修正が必要 p.1, p.6

NB. 3*3 Magic Square ==============================================
N =: i.3
M0 =: (3,3) $ N
MR =: (N ,: (1 |. N)) , (2 |. N)
ML =: (N ,: (2 |. N)) , (1 |. N)

Mahoujin =: 3 3 $ ; >: (L:0) 3 #. L:0 (< "(0) ML) ,L:0 (< "(0) MR)

Magic =: ((2{Mahoujin),: (0{Mahoujin)) , (1{Mahoujin)

diag =: i. @ # {"_1 ]


NB. 5*5 Magic Square ==============================================

NN =: i.5
MM0 =: (5,5) $ NN
MMR =: ((NN ,: (1 |. NN)) , (2 |. NN) ,: (3 |. NN)) , (4 |. NN)
MML =: (((NN ,: (4 |. NN)) , (3 |. NN)), (2 |. NN)) , (1 |. NN)

MMahoujin =: (<"(0) MML) , L:0  (<"(0) MMR)

MMagic =: (5, 5) $ ; 5 #. L:0 MMahoujin

Magic55 =: 1 + ((3, 4) { MMagic) ,  (0, 1, 2) { MMagic
```